content: -- Copyright (c) 2016-present, Facebook, Inc.
  -- All rights reserved.
  --
  -- This source code is licensed under the BSD-style license found in the
  -- LICENSE file in the root directory of this source tree.
  
  {-# LANGUAGE GADTs #-}
  {-# LANGUAGE LambdaCase #-}
  {-# LANGUAGE OverloadedStrings #-}
  
  module Duckling.Duration.TR.Rules
    ( rules
    ) where
  
  import Control.Monad (join)
  import Data.String
  import Prelude
  import qualified Data.Text as Text
  
  import Duckling.Dimensions.Types
  import Duckling.Duration.Helpers
  import Duckling.Numeral.Helpers (parseInteger)
  import Duckling.Numeral.Types (NumeralData (..))
  import Duckling.Regex.Types
  import Duckling.Types
  import qualified Duckling.Numeral.Types as TNumeral
  import qualified Duckling.TimeGrain.Types as TG
  
  ruleDurationQuarterOfAnHour :: Rule
  ruleDurationQuarterOfAnHour = Rule
    { name = "quarter of an hour"
    , pattern = [ regex "(1/4\\s?sa(at)?|çeyrek saat)" ]
    , prod = \_ -> Just . Token Duration $ duration TG.Minute 15
    }
  
  ruleDurationHalfAnHour :: Rule
  ruleDurationHalfAnHour = Rule
    { name = "half an hour"
    , pattern = [regex "(1/2\\s?sa(at)?|yar\305m saat)"]
    , prod = \_ -> Just . Token Duration $ duration TG.Minute 30
    }
  
  ruleDurationThreeQuartersOfAnHour :: Rule
  ruleDurationThreeQuartersOfAnHour = Rule
    { name = "three-quarters of an hour"
    , pattern = [regex "(3/4\\s?sa(at)?|üç \231eyrek sa(at)?)"]
    , prod = \_ -> Just . Token Duration $ duration TG.Minute 45
    }
  
  ruleDurationFortnight :: Rule
  ruleDurationFortnight = Rule
    { name = "fortnight"
    , pattern = [regex "iki hafta"]
    , prod = \_ -> Just . Token Duration $ duration TG.Day 14
    }
  
  ruleNumeralQuotes :: Rule
  ruleNumeralQuotes = Rule
    { name = "<integer> + '\""
    , pattern =
      [ Predicate isNatural
      , regex "(['\"])"
      ]
    , prod = \tokens -> case tokens of
        (Token Numeral NumeralData{TNumeral.value = v}:
         Token RegexMatch (GroupMatch (x:_)):
         _) -> case x of
           "'"  -> Just . Token Duration . duration TG.Minute $ floor v
           "\"" -> Just . Token Duration . duration TG.Second $ floor v
           _    -> Nothing
        _ -> Nothing
    }
  
  ruleDurationDotNumeralHours :: Rule
  ruleDurationDotNumeralHours = Rule
    { name = "number.number hours"
    , pattern = [regex "(\\d+)(\\.|,)(\\d+) *sa(at)?"]
    , prod = \tokens -> case tokens of
        (Token RegexMatch (GroupMatch (h:m:_)):_) -> do
          hh <- parseInteger h
          mnum <- parseInteger m
          let mden = 10 ^ Text.length m
          Just . Token Duration $ minutesFromHourMixedFraction hh mnum mden
        _ -> Nothing
    }
  
  ruleDurationAndHalfHour :: Rule
  ruleDurationAndHalfHour = Rule
    { name = "<integer> and an half hour"
    , pattern =
      [ Predicate isNatural
      , regex "buçuk sa(at)?"
      ]
    , prod = \tokens -> case tokens of
        (Token Numeral NumeralData{TNumeral.value = v}:_) ->
          Just . Token Duration . duration TG.Minute $ 30 + 60 * floor v
        _ -> Nothing
    }
  
  
  ruleDurationPrecision :: Rule
  ruleDurationPrecision = Rule
    { name = "<duration> about|exactly"
    , pattern =
      [ regex "(yaklaşık|tam(\\solarak)?)"
      , dimension Duration
      ]
      , prod = \tokens -> case tokens of
          (_:token:_) -> Just token
          _           -> Nothing
    }
  
  ruleDurationPrecision2 :: Rule
  ruleDurationPrecision2 = Rule
    { name = "<duration> about|exactly"
    , pattern =
      [ dimension Duration
      , regex "gibi|civarında"
      ]
    , prod = \case
        (token:_:_) -> Just token
        _           -> Nothing
    }
  
  rules :: [Rule]
  rules =
    [ ruleDurationQuarterOfAnHour
    , ruleDurationHalfAnHour
    , ruleDurationThreeQuartersOfAnHour
    , ruleDurationFortnight
    , ruleDurationDotNumeralHours
    , ruleDurationAndHalfHour
    , ruleDurationPrecision
    , ruleDurationPrecision2
    , ruleNumeralQuotes
    ]
  